Pomůcka pro přednášku: 3. semestr Bc studia

Pravděpodobnost a statistika - dvourozměrná náhodná proměnná a její charakteristiky

Dvourozměrná NP, určit $a,\ EX, EY, \rho(X,Y)$.

Př. Hodnoty X, Y dvourozměrné NP diskrétního typu mohou nabývat hodnot $X=(1,2,3)$ resp. $Y=(-2,-1,0,1,2)$. Hodnoty simultánní pravděpodobnostní funkce $P(X=i, Y=j)$ jsou zadány maticí $A=\begin{bmatrix} a & 0.07 & 0.03 & 0.01 & 0.05 \\ 0.05 & 0.07 & a & 0.2 & \frac{a}{2} \\ 0.05 & 0.06 & 0.07 & 0.04 & \frac{a}{2} \end{bmatrix}$. Určete hodnotu a a vypočtěte střední hodnoty $E(X), E(Y)$ a koeficient korelace $\varrho(X,Y)$.

```
> X:=[1,2,3]; Y:=[-2,-1,0,1,2];
A:=Matrix(3,5,[a,0.07,0.03,0.01,0.05,0.05,0.07,a,0.2,a/2,0.05,0.06,.07,0.04,a/2]);
```

$$X := [1, 2, 3]$$

$$Y := [-2, -1, 0, 1, 2]$$

$$A := \begin{bmatrix} a & 0.07 & 0.03 & 0.01 & 0.05 \\ 0.05 & 0.07 & a & 0.2 & \frac{1}{2}a \\ 0.05 & 0.06 & 0.07 & 0.04 & \frac{1}{2}a \end{bmatrix}$$

```
> soucet:=add(i,i=A);
```

$$soucet := 0.70 + 3\,a$$

```
> a:=solve(soucet=1,a);
```

$$a := 0.1000000000$$

```
> ii:=[op(1,A)][1]; jj:=[op(1,A)][2];
```

$$ii := 3$$

$$jj := 5$$

```
> P_X:=[seq(add(A[i,j],j=1..jj),i=1..ii)];
```

$$P\_X := [0.2600000000, 0.4700000000, 0.2700000000]$$

```
> P_Y:=[seq(add(A[i,j],i=1..ii),j=1..jj)];
```

$$P\_Y := [0.2000000000, 0.20, 0.2000000000, 0.25, 0.1500000000]$$

```
> EX:=add(X[i]*P_X[i],i=1..ii);
```

$$EX := 2.010000000$$

```
> EY:=add(Y[j]*P_Y[j],j=1..jj);
```

$$EY := -0.0500000000$$

```
> DX:=add(X[i]^2*P_X[i],i=1..ii)-EX^2;
sigma_X:=sqrt(DX);
```

$$DX := 0.529900000$$

$$sigma\_X := 0.7279423054$$

```
> DY:=add(Y[j]^2*P_Y[j],j=1..jj)-EY^2;
sigma_Y:=sqrt(DY);
```

$$DY := 1.847500000$$

$$sigma\_Y := 1.359227722$$

```
> C_XY:=add(add(X[i]*Y[j]*A[i][j],j=1..jj),i=1..ii)-EX*EY;
```

$$C\_XY := 0.1405000000$$

```
> rho_XY:=C_XY/(sigma_X*sigma_Y);
```

$$rho\_XY := 0.1419996065$$